
Abstract

Procedurally reective programming languages enable user
programs to semantically extend the language itself, by per-
mitting them to run at the level of the language implementa-
tion with access to their context. The reective tower, �rst
introduced by Brian Smith [9, 10], is the principal archi-
tecture for such languages. It is informally described as an
in�nitely ascending tower of meta-circular interpreters, con-
nected by a mechanism that allows programs at one level to
run at the next higher level. Various accounts of the reec-
tive tower have been published, including a meta-circular
de�nition, operational de�nitions, and denotational de�ni-
tions. We present an operational account of the main as-
pects of the reective tower, which we claim is simpler than
previous accounts. Our approach is to implement a �nite
tower where each level literally runs the level directly below
it. A complete Scheme implementation is included.

1 Introduction

The principal characteristic of reective programming lan-
guages is that they are extensible. That is, they permit the
user to de�ne new language constructs at the semantic level
by e�ectively adding lines of code to the language processor
itself. Lisp-based reective programming languages, such as
3-Lisp [9, 10], Brown [12], and Blond [2], provide such an
extension capability by permitting user-level code to be run
as if it were interpreter code with access to the interpreter's
current expression, environment, and continuation. With
this language extension capability, otherwise primitive lan-
guage constructs, such as catch, throw, call/cc, boundp, and
quote, can be de�ned as user procedures.

The reective tower, �rst introduced by Smith [9, 10],
provides a uni�ed and coherent architecture for reec-
tive programming languages. The reective tower is non-
e�ectively described as an in�nitely ascending tower of meta-
circular interpreters. The interpreter at the bottom of the

tower executes user input, and every other interpreter in the
tower executes the interpreter immediately below it. The
levels are crucially connected by a mechanism that permits
a program running at one level to provide code that is to run
at the next higher level. Various accounts of the reective
tower architecture have been given including a meta-circular
de�nition [9, 10], operational de�nitions [4, 1], and denota-
tional de�nitions based on the meta-continuation concept
[12, 2]. We give another account that we believe is simpler
and more easily understood than previous accounts.

Our approach is to give a literal model of a �nite reec-
tive tower. An interpreter for a small, simpli�ed subset of
Scheme [8] is implemented in Scheme. This interpreter is
written in its own subset of Scheme so that it can literally
run itself. This permits �nite towers of the interpreter to
be run (albeit ine�ciently). Then, a few modi�cations are
made so that the interpreters in a �nite tower are connected
as in the in�nite reective tower. Once the literal model of a
�nite reective tower is understood, the behavior of an in�-
nite reective tower can be easily understood as the limiting
behavior of a �nite reective tower as its height approaches
in�nity.

The literal de�nition is similar to the meta-circular de�-
nition of 3-Lisp. It di�ers in that it runs, it is not circular,
and it does not reconstruct evaluation and quotation. The
literal model is simpler than the operational and denota-
tional models since it models the reective tower implicitly
rather than explicitly. All of the models are interesting and
useful because they clarify di�erent aspects of reection.

The rest of this paper is organized as follows: section 2
gives a brief overview of the reective tower architecture;
section 3 presents a simple interpreter that is capable of
interpreting itself; section 4 derives a language extension
mechanism for this interpreter; section 5 presents the imple-
mentation of our reective interpreter; section 6 is a com-
parison with related work; and section 7 is the conclusion.

2 Overview of the Reective Tower Architecture

The reective tower is described as an in�nitely ascending
tower of meta-circular interpreters. The interpreter at the
tower's bottom executes user input, and every other inter-
preter in the tower executes the interpreter immediately be-
low it. Each interpreter state is passed around as three ar-
guments to the interpreter: an expression, environment and
continuation. An interpreter running at one level may have
a di�erent state than an interpreter running at a di�erent
level. Normally, the interpreters share a global environment.

Reifying procedures provide a mechanism for running
code at the next higher level and gaining access to that
level's state in the form of an expression, environment, and
continuation. A reifying procedure, or simply rei�er, has
three formal parameters. When a rei�er is applied, its body
is run as if it were code belonging to the interpreter run-
ning the application. The body is run in an environment
where the �rst formal parameter is bound to the list of un-
evaluated arguments, the second formal parameter is bound
to the environment argument of the interpreter running the
application, and the third formal parameter is bound to the
continuation argument of the interpreter. In a fully meta-
circular model of the architecture, the body of a rei�er has
access to the procedures and structures that the interpreter
itself has. Since reifying procedures can themselves invoke
reifying procedures, it is possible to run user code at arbi-
trarily high levels of the tower.

3 A Simple Interpreter

We begin by presenting a garden variety interpreter for a
subset of Scheme. This interpreter, referred to as I, is writ-
ten in its own language so that it can interpret itself. It
is written in continuation-passing style. The extended BNF
for the subset of the Scheme language implemented by I fol-
lows (non-terminals are printed in italic font and terminals
are printed in typewriter font).

exp ::= (exp fexpg�)

j (lambda (fidentifierg�) fexpg+)

j (if exp exp exp)

j (set! identifier exp)

j (quote exp)

j identifier

j constant

The syntactic categories of identi�ers and constants are the
same as in Scheme. The remaining notable characteristics
of I follow: set! assigns to an unbound identi�er; only a
small subset of Scheme primitives is implemented; and error
detection and reporting is almost non-existent.

Other than expressions, the principal structures manipu-
lated by I are environments, continuations, and procedures.
As usual, a continuation is a procedure of one argument.
Environments are represented by association lists. Each ele-
ment of the association list is a pair consisting of an identi�er
and a value. The interpreter handles two types of proce-
dures. The two types are represented as lists and are dis-
tinguished by a tag. A compound procedure is represented
by a list containing a tag, formal parameters, a procedure
body, and an environment. The body of a compound proce-
dure is a non-empty list of expressions. The environments
contained in compound procedures and passed around by
the interpreter only list the bindings that extend the global
environment; whenever a variable binding is looked up (via
get-pair) the global environment is searched if necessary. A
primitive procedure contains a tag and an operator symbol.
For example, the identity procedure is '(compound (x) (x)
()), and the addition procedure is '(primitive +). We
follow the convention that e, r and k are expression, envi-
ronment, and continuation, respectively.

Following are the Scheme de�nitions of some of the prin-
cipal procedures that de�ne I. Since the interpreter must
interpret itself, the de�nitions are also I de�nitions. All of
the supporting de�nitions are given in the appendix. The
procedure evaluate dispatches on the syntactic type of its
expression operand e.

(set! evaluate
(lambda (e r k)

((if (constant? e)
evaluate-constant
(if (variable? e)
evaluate-variable
(if (if? e)
evaluate-if
(if (assignment? e)
evaluate-assignment
(if (abstraction? e)
evaluate-abstraction
evaluate-combination)))))

e r k)))

As usual, an abstraction simply evaluates to a procedure.
Following are the details, since a good understanding is re-
quired in the next section.

(set! evaluate-abstraction
(lambda (e r k)

(k (make-compound
(formals-part e) (body-part e) r))))

(set! make-compound
(lambda (formals body r)

(list 'compound formals body r)))

Combinations are evaluated in a conventional manner.
The operator is evaluated to obtain a procedure, and the
operands are evaluated to obtain arguments. Then the body
of the procedure is evaluated in the procedure's lexical en-
vironment extended by binding the procedure's formal pa-
rameters to its arguments.

(set! evaluate-combination
(lambda (e r k)

(evaluate (operator-part e) r
(lambda (proc)
(evaluate-operands (operands-part e) r
(lambda (args)
(apply-procedure proc args k)))))))

(set! apply-procedure
(lambda (proc args k)

(if (compound? proc)
(evaluate-sequence
(procedure-body proc)
(extend
(procedure-environment proc)
(procedure-parameters proc)
args)

k)
(k (apply-primitive

(procedure-name proc) args)))))

Evaluate-operands produces a list of arguments from a list
of operands. Evaluate-sequence successively evaluates a

sequence of expressions, yielding the value of the �nal ex-
pression. Apply-primitive simply passes the task of ap-
plying a primitive to the language processor running the
interpreter.

The global environment of I, bound to global-env,must
be initialized with bindings for the primitive procedures.

(set! initialize-global-env
(lambda ()
(set! global-env
(extend
empty-env
(primitive-identifiers)
(mapper make-primitive
(primitive-identifiers))))))

A read-eval-print loop is provided by the following pro-
cedure.

(set! openloop
(lambda (read-prompt write-prompt)

(display read-prompt)
(evaluate (read) empty-env
(lambda (v)
(display write-prompt)
(if (eq? v (void))

"Nothing will be displayed"
(write v))

(newline)
(openloop read-prompt write-prompt)))))

Openloop always prints its output prompt after the input
expression has �nished running, even though no output is
printed when an expression returns a void value.

The following transcript illustrates starting the inter-
preter and running an expression (\>" is the Scheme prompt).

> (load "simple.ss")
> (initialize-global-env)
> (openloop "0> " "0: ")
0> (+ 2 3)
0: 5
0>

Since the interpreter is written in its own language, it is
possible to load the interpreter and have it run itself. The
de�nition of I includes the following procedure for loading
source �les.

(set! loadfile
(lambda (file)
((lambda(port)

((lambda (loop)
(set! loop
(lambda (v)
(if (eof-object? v)
(close-input-port port)
(evaluate v empty-env
(lambda (ignore)
(loop (read port)))))))

(loop (read port)))
'*))

(open-input-file file))))

Using, loadfile, we create a two-level tower of I inter-
preters:

> (load "simple.ss")
> (initialize-global-env)
> (loadfile "simple.ss")
> (openloop "0> " "0: ")
0> (initialize-global-env)
0:
0> (openloop "1> " "1: ")
1> (+ 2 3)
1: 5
1>

The absence of a level-0 output prompt after the second
openloop indicates that the second openloop has not �n-
ished running. We orient the tower in the same way as
3-LISP (i.e., the level above runs the level below), but our
level numbering convention is reversed from that of 3-LISP.

I traps unbound variable errors and exits. Notice how
a level is lost in the following continuation of the previous
transcript.

1> xxx
Error: symbol not bound: xxx
0:
0>

The level-1 I displays the error message and exits. Control
then passes to the level-0 openloop, which indicates that
level-1 returned a void value by printing a level-0 output
prompt with no value following it.

4 Adding an Extension Mechanism

We now consider adding a mechanism to I that would en-
able user-supplied code to be run at the level of the inter-
preter. As it stands, I never runs user code at the level of
the interpreter. Rather, user code is simply inspected, de-
composed and manipulated by the code that implements I.
The mechanism that we propose to study is a (rather im-
practical) special form (exit exp) that runs its unevaluated
operand at the level of the interpreter code and then exits
the interpreter. We've chosen exit solely for its simplicity
and ability to illustrate some fundamental ideas. We use
Iexit to refer to the putative interpreter that extends I with
the exit mechanism. The following illustrates the behavior
of exit in a two-level tower.

1> (+ 5 (* 3 2))
1: 11
1> (+ 5 (exit (* 3 2)))
0: 6
0>

At the very least, a new line must be added to evaluate
in order to dispatch exit forms. Assume that in Iexit the
following line has been added to the de�nition of evaluate
immediately after the dispatch line for assignment (and that
parentheses have been appropriately adjusted).

(if (exit? e) evaluate-exit

Then, the evaluation of (exit exp) should proceed as if
evaluate-exit were temporarily de�ned as

(lambda (e r k)
exp)

We need a way of abstracting exp from this expression in
terms of e. This is not directly possible since lambda does
not evaluate its arguments. Also, we reject as too compli-
cated and inelegant the possibility of dynamically modify-
ing the procedure objects constituting Iexit. We proceed
by deriving an acceptable de�nition of evaluate-exit from
the previous lambda expression. The derivation requires the
additional assumption that exp is closed in the global envi-
ronment. The preceding lambda expression can be written
as

(lambda (e r k)
((lambda () exp)))

Since a lambda expression evaluates to a procedure (formed
by make-compound), we may write the above as

(lambda (e r k)
((make-compound

'() (list (quote exp)) '())))

Finally, we may abstract exp, yielding the following de�ni-
tion of evaluate-exit

(set! evaluate-exit
(lambda (e r k)
((make-compound

'() (cdr e) '()))))

The expression

(make-compound '() (cdr e) '())

evaluates to a procedure object, say proc, of Iexit. The
problem is that proc probably won't be recognized as a pro-
cedure object by the language processor running Iexit. For
example, when only one level of Iexit is run by our Scheme
system, the following session ensues:

0> (exit (* 3 2))
Error: attempt to apply non-procedure

(compound () ((* 3 2)) ()).
>

However, if the language processor that is running Iexit rec-
ognizes the procedure objects of Iexit, then this de�nition of
evaluate-exit works. Thus, we are led to consider a two-
level tower of interpreters, where Iexit is run by either Iexit
or I. The more levels there are in the tower, the higher in
the tower an exit expression can be successfully run. For
example, the expression (exit (exit 17)) fails in a two-
level tower:

0> (openloop "1> " "1: ")
1> (exit (exit 17))
Error: attempt to apply non-procedure

(compound () (17) ()).
>

On the other hand, (exit (exit 17)) would run without
error in a three-level tower, returning to level 0. Of course,
uniformity of the levels would be achieved by modeling an
in�nite tower as in Brown, 3-LISP, and Blond.

There is nothing mysterious about our �nal de�nition of
evaluate-exit. The expression

((make-compound '() (cdr e) '()))

in the de�nition of evaluate-exit runs the argument of
exit at the level of interpreter code because it is an instance
of the following basic observation:

If the combination (exp0 exp1 :::) is run at level
l and exp0 evaluates to a compound procedure
object p, then the body of p is run at level l.

The ability to run user code at the level of interpreter code
is a consequence of:

1. the basic observation;

2. the capability for user code to create procedure objects
dynamically; and

3. the capability of the interpreter to interpret itself, so
that a tower could be created.

Item 2 is included because an interpreter running at level
l + 1 is run as user code by the interpreter at level l (using
our level numbering for prompts).

In Iexit we have an instance of crossing semantic levels,
where exp is treated both as data and code. When Iexit
interprets the user expression (exit exp), it treats the ex-
pression as data: extracting its car, dispatching on its car,
passing it as an argument, extracting its cdr, embedding its
cdr in a procedure object. But, then the procedure object
with exp in its body is applied as interpreter code, and exp
is run as interpreter code by the language processor running
Iexit. In contrast, I always treats user expressions as data
and never allows user expressions to be run as interpreter
code.

The reader may be wondering why we couldn't just de-
�ne evaluate-exit as follows, and avoid the whole issue of
towers.

(set! evaluate-exit
(lambda (e r k)

(evaluate (2nd e) r (lambda (x) x))))

This de�nition does not satisfy our requirement that the
argument to exit run at the same level as interpreter code.
Indeed, with this de�nition, the argument to exit is always
treated as data by the interpreter. The following transcript,
when compared with the previous transcript, illustrates a
behavioral di�erence between this de�nition and our derived
de�nition.

0> (openloop "1> " "1: ")
1> (exit (exit 17))
0: 17
0>

5 A Reective Interpreter

We now turn our attention to de�ning a reective interpreter
based on a �nite, literal tower. First, we extend the de�ni-
tion of I in order to solve a couple of technical problems
with literal towers. Then, we make some extensions and a
minor modi�cation so that the interpreter can handle reify-
ing procedures. We refer to the resulting interpreter as IR.
A complete de�nition of IR, including all utility functions,
can be found in the appendix.

5.1 A Self-interpreting Interpreter

The method of building towers in the previous sections has
two drawbacks. The �rst drawback is that the global envi-
ronment is not structurally shared among the di�erent levels
of a tower. The global environment at each level of a tower

is initialized with the same de�nitions, but the global envi-
ronment of each level is entirely constructed using new pairs.
Hence, assignments at one level are not observable at other
levels. This is illustrated by the following transcript.

> (load "simple.ss")
> (initialize-global-env)
> (loadfile "simple.ss")
> (openloop "0> " "0: ")
0> (set! level-0-var 0)
0:
0> level-0-var
0: 0
0> (initialize-global-env)
0:
0> (openloop "1> " "1: ")
1> (set! level-1-var 1)
1:
1> level-1-var
1: 1
1> level-0-var
Error: symbol not bound: level-0-var
0:
0> level-1-var
Error: symbol not bound: level-1-var
>

Our solution to this inconvenience is to simply share the
global environment among all levels. The second drawback
is that the �le containing the interpreter de�nition must be
loaded each time a new level is created. Since every level
results in an order of magnitude slowdown, the time that it
takes to create levels quickly becomes unbearably long, and
having to wait inhibits experimentation. Sharing the global
environment between levels makes it unnecessary to load the
interpreter de�nition each time a new level is created.

Let's look more closely at the environment structure in
a tower. Consider the previous transcript. The �rst line

> (load "simple.ss")

loads the procedure de�nitions for the interpreter into the
Scheme environment. Then, the global environment of the
interpreter is initialized to have bindings for the primitive
procedures.

> (initialize-global-env)

In order to run a level-1 openloop, the level-0 interpreter's
global environment must also include the de�nitions for the
interpreter. This is accomplished by reloading the same �le
into the interpreter's global environment.

> (loadfile "simple.ss")

Then we start the interpreter's read-eval-print loop.

> (openloop "0> " "0: ")
0>

Although we could now open another level, it would fail
because its global environment global-env is not bound in
the level-0 interpreter's environment.

0> (openloop "1> " "1: ")
1> (+ 1 2)
Error: symbol not bound: global-env
>

The level-0 interpreter was attempting to run level-1 inter-
preter code that contained a reference to global-env. It was
the level-0 interpreter that determined that global-envhad
not been bound, and thus exited to Scheme.

We can share the global environment of the interpreter
among levels by including a binding for global-env in the
level-0 interpreter's global environment. This binding must
reference the interpreter's global environment so that all
modi�cations are shared among all levels. Thus the global
environment of the interpreter becomes a self-referential, cir-
cular structure. The following procedure initializes the in-
terpreter's global environment so that it is shared among all
levels, and starts a read-eval-print loop.

(set! boot-tower
(lambda ()

(initialize-global-env)
(loadfile this-file-name)
(set-cdr! global-env
(cons (cons 'global-env global-env)

(cdr global-env)))
(openloop "0> " "0: ")))

With an initialized, self-referential, global environment,
it is no longer necessary to run initialize-global-env and
loadfile each time a new level is created. The following
transcript creates a tower with three levels.

> (load "simple.ss")
> (boot-tower)
0> (openloop "1> " "1: ")
1> (openloop "2> " "2: ")
2>

Each successive openloop still results in an order of mag-
nitude slowdown, but simply starting a new level is much
faster than the method of previous sections.

We also have the capability to call evaluate or any other
implementation procedure:

2> (evaluate '(+ 1 x) '((x . 3))
(lambda (v) (* v 11)))

2: 44
2>

5.2 Reifying Procedures

We now consider adding reifying procedures to the inter-
preter. Reifying procedures are the mechanism that 3-LISP,
Brown, and Blond provide to enable user code to run at
the level of the interpreter with access to the interpreter's
current expression, environment, and continuation.

5.2.1 Overview

In IR, reifying procedures are created by applying the proce-
dure compound-to-reifier to a three-argument compound
procedure. The result is a three-argument reifying proce-
dure. When the operator of a combination evaluates to
a reifying procedure, the body of the reifying procedure
is run as if it were code of the interpreter evaluating the
combination; the body is run in an extension of the reify-
ing procedure's lexical environment where the �rst formal
parameter is bound to the list of unevaluated operands of
the combination, the second formal parameter is bound to
the environment argument of the interpreter evaluating the

combination, and the third formal parameter is bound to
the continuation argument of the interpreter evaluating the
combination. To illustrate, a transcript follows showing the
de�nition and use of a reifying procedure that aborts the
current level, returning its unevaluated �rst argument.

2> (set! quit
(compound-to-reifier
(lambda (e r k) (1st e))))

2:
2> (quit foo)
1: foo
1> (quit bar)
0: bar
0>

Interesting reifying procedures typically call evaluate
in their bodies. Following is the de�nition of a special form
(when test exp) that evaluates exp if and only if test eval-
uates to true.

2> (set! when
(compound-to-reifier
(lambda (e r k)
(evaluate (1st e) r
(lambda (v)
(if v

(evaluate (2nd e) r k)
(k (void))))))))

2:
2> (when (= 1 2) "Shouldn't happen")
2:
2>

It is useful to de�ne a special form alpha for abstracting
rei�ers. With alpha, quit could be de�ned by

(set! quit (alpha (e r k) (1st e)))

The body of an alpha is closed in the environment of its def-
inition. The de�nition of alpha uses make-reifier, which,
like make-compound, takes a list of formal parameters, a pro-
cedure body, and an environment, and directly creates the
associated reifying procedure. We de�ne alpha as follows:

(set! alpha
(compound-to-reifier

(lambda (e r k)
(k (make-reifier (1st e) (cdr e) r)))))

5.2.2 Implementation

Reifying procedures are represented like compound proce-
dures, except that the tag is 'reifier instead of 'compound.
Converting between reifying procedures and compound pro-
cedures is then simply a matter of changing tags. Following
are the operations on reifying procedures.

(set! make-reifier
(lambda (formals body r)
(list 'reifier formals body r)))

(set! reifier-to-compound
(lambda (reifier)
(cons 'compound (cdr reifier))))

(set! compound-to-reifier
(lambda (compound)
(cons 'reifier (cdr compound))))

All that remains is to handle the evaluation of combi-
nations where the operator evaluates to a reifying proce-
dure. Suppose that we have a combination (exp0 exp1 :::)
where exp0 evaluates to a reifying procedure bound to proc.
Let ext be the extension of the lexical environment of proc
where the formal parameters of proc are bound to the list
of unevaluated operands of the combination, the environ-
ment of the interpreter evaluating the combination, and the
continuation of the interpreter evaluating the combination,
respectively. We want to run the body of proc at the level
of interpreter code in the environment ext. This is the same
as running the body of the value of

(reifier-to-compound proc)

at the level of interpreter code in the environment ext. By
the basic observation, this is the same as running

((reifier-to-compound proc) (operands-part e) r k)

at the level of interpreter code. Thus, the following modi�-
cation to evaluate-combination, indicated by semicolons,
runs the body of a rei�er at the proper level.

(set! evaluate-combination
(lambda (e r k)

(evaluate (operator-part e) r
(lambda (proc)
(if (reifier? proc) ;
((reifier-to-compound proc) ;
(operands-part e) r k) ;

(evaluate-operands (operands-part e) r
(lambda (args)

(apply-procedure proc args k))))))))

This completes the description of the changes made to I in
order to obtain IR. The full de�nition of IR appears in the
appendix.

The expression

((reifier-to-compound proc) (operands-part e) r k)

is the essence of our reective tower model. It is essentially
identical to the corresponding expression

(#(de-reflect proc!) args env cont)

in the meta-circular de�nition of 3-Lisp [10]. The 3-Lisp
procedure de-reflectperforms the same function as the IR
procedure reifier-to-compound: it converts the closure for
a rei�er into a closure for a compound procedure. The only
signi�cant di�erence between the two expressions is the use
of the meta-structural operator # in the 3-Lisp expression
so that the operator of the combination is a program (des-
ignating a function) rather than a structure (designating a
closure).

5.3 Observations

The reective interpreter IR was obtained by making a few
simple modi�cations to I. The �rst modi�cation created a
self-referential global environment so that the interpreters
in a tower could share the global environment. Then a
datatype for reifying procedures was de�ned and a case was
added to evaluate-combination so that it could handle ap-
plications of rei�ers.

An error from the underlying Scheme system usually re-
sults if code run by level-0 attempts to apply a rei�er:

0> ((compound-to-reifier
(lambda (e r k) "Wow!")))

Error: attempt to apply non-procedure
(compound (e r k) ("Wow!") ()).

>

The error is due to running the expression

((reifier-to-compound proc) (operands-part e) r k)

as level-0 IR interpreter code. The level-0 IR is run by the
underlying Scheme system, which does not understand the
representation of IR procedures.

The following illustrates how it is possible to climb out
of the �nite tower and into Scheme, resulting in an error:

0> (openloop "1> " "1: ")
1> (openloop "2> " "2: ")
2> (set! climb

(compound-to-reifier
(lambda (e r k)
(evaluate (1st e) r
(lambda (v)
(if (= v 0)
(k v)
(climb (- v 1))))))))

2:
2> (climb 0)
2: 0
2> (climb 1)
1: 0
1> (openloop "2> " "2: ")
2> (climb 2)
Error: attempt to apply non-procedure

(compound (e r k) ((evaluate ...)) ()).
>

Due to the meta-circularity of IR, user programs have
full access to the implementation and structures of IR. This
is illustrated in the following transcript.

0> (openloop "1> " "1: ")
1> (openloop "2> " "2: ")
2> ((compound-to-reifier

(lambda (e r k)
(list e r k)))

(* 3 5 ))
1: (((* 3 5))

()
(compound (v)
((display write-prompt)
(if (eq? v (void))
"Nothing will be displayed"
(write v))

(newline)
(openloop read-prompt write-prompt))

((write-prompt . "2: ")
(read-prompt . "2> "))))

1> evaluate-abstraction
1: (compound (e r k)

((k (make-compound
(formals-part e)
(body-part e)
r)))

())
1>

6 Comparison with Related Work

3-Lisp, Brown, Blond, and IR are all closely related. All
are interpreters based on the reective tower architecture
described in section 2. IR models a �nite reective tower,
whereas the others model an in�nite reective tower. All
four languages are lexically scoped, higher order, applica-
tive order dialects of Lisp. We do not include Stepper[1]
because its reective architecture di�ers substantially from
that considered here.

3-Lisp di�ers in signi�cant ways from other Lisp dialects.
Before reection is introduced, a semantically rationalized
Lisp is de�ned. New internal structures are introduced in
order to uniquely represent semantic concepts (e.g., combi-
nations are not represented by list structure) and syntax is
assigned to uniquely represent each type of internal struc-
ture. This results in an alignment of syntactic, structural,
and semantic concepts. Quotation is a structural primitive
in 3-Lisp that is used to designate internal structures. Lev-
els of designation can only be removed or added explicitly
via up and down. Evaluation is replaced by normalization,
which takes an expression into a co-designating structure in
normal form (e.g., '2 normalizes to '2, not 2). The result is
that semantical levels may only be crossed with the explicit
use of up and down. Reection in 3-Lisp is then described as
in section 2.

Meta-circular de�nitions of 3-Lisp are given in [9, 10, 11].
The meta-circular de�nition uses 3-Lisp to de�ne 3-Lisp and
hence is not well-founded. The de�nition of IR is similar
to the meta-circular de�nition of 3-Lisp. Level shifting and
meta-circular access are implicit in both de�nitions, and the
code for applying rei�ers is essentially identical. The de�ni-
tion of IR di�ers in the following respects:

� It models a �nite tower, rather than an in�nite tower.

� It is well-founded and can be run in Scheme. The
meta-circular de�nition of 3-Lisp can only be run by a
3-Lisp implementation.

� It does not require an understanding of 3-Lisp's recon-
struction of evaluation and quotation.

The non-reective implementations [9, 4] of 3-Lisp are
extremely operational. In order to operate e�ciently, level
shifting in the tower is explicitly modeled with a stack of
continuations corresponding to the active levels of the tower.
Full access to the interpreter, implied by meta-circularity, is
explicitly simulated.

Brown and Blond are reective dialects of Scheme, both
of which are implemented in Scheme. Both are essentially
executable denotational de�nitions of the non-meta-circular
aspects of the reective tower architecture. Level shifting
in the tower is explicitly modeled by adding a new context
argument, called the meta-continuation, to the semantic val-
uation functions.

As IR demonstrates, meta-circularity implies that user
programs have the same access privileges as the interpreter
itself. Thus, the procedures that implement a meta-circular
interpreter may be used, inspected, and modi�ed by a user
program. Environments and continuations can be similarly
accessed by a user program. IR user programs have the
same access privileges as IR itself. The implementation of
3-Lisp faithfully simulates its corresponding meta-circular
de�nition, but protects itself by not allowing modi�cation

of the standard 3-Lisp system. Nevertheless, 3-Lisp user
programs have full inspection privileges.

In Blond and Brown, the internal structure of the in-
terpreter is largely hidden from user programs. Brown and
Blond make no attempt to model the meta-circular prop-
erties of the reective tower, but rather focus on the deno-
tational semantics of shifting between levels of the tower.
Both Brown and Blond rely upon Scheme closures to rep-
resent user-de�ned procedures, interpreter procedures and
interpreter continuations. Brown also represents the envi-
ronment as a closure. The only sanctioned operation that
either the Blond or Brown interpreters or their user pro-
grams can perform on Scheme closures is application.

In 3-Lisp, Brown, and IR, rei�ers close in the environ-
ment of their de�nition. In Blond, rei�ers close in either
the environment of the level above their de�nition or in the
environment above their level of application. Rei�ers of the
�rst kind are de�ned via gamma abstractions, and those of
the second kind are de�ned via delta abstractions.

In 3-Lisp, Brown, and IR, if the application of a rei�ed
continuation (i.e., interpreter continuation) returns, then
execution is resumed at the point following the applica-
tion. This procedure-like behavior of rei�ed continuations
has been termed \pushy" [5]. In IR this pushy behavior is
a natural consequence of actually running a literal tower of
interpreters. Brown and 3-Lisp simulate the pushy behavior
that is a consequence of the reective tower architecture.
The designers of Blond argue for \jumpy" rei�ed continua-
tions, which behave more like a goto when they are applied.
The jumpy behavior is obtained by forgetting the continua-
tion in e�ect when a rei�ed continuation is applied.

Blond has a separate environment on each level and a
global environment that is common to all levels. Brown and
IR have only a global environment that is common to all
levels. Both approaches have been used in di�erent 3-Lisp
implementations.

7 Conclusion

An executable reective interpreter has been de�ned by mak-
ing a fewminor modi�cations to an ordinary interpreter for a
dialect of Scheme. The reective interpreter literally models
a �nite reective tower. This makes it too slow for practical
use, but its simplicity makes it ideal for understanding and
experimenting. Once the literal model of a �nite reective
tower is understood, the behavior of an in�nite reective
tower is easily understood as the limiting behavior of a �-
nite reective tower as its height approaches in�nity.

Our reective interpreter illustrates how switching levels
in the tower involves treating data as code and vice-versa.
For those who are uncomfortable with this sort of semantical
level crossing, it serves as additional motivation for under-
standing the reconstruction of evaluation and quotation in
3-Lisp. [9, 10].

Appendix: The Complete Reective Interpreter

Following is a listing of a Scheme source �le, which we have
named simple.ss, that de�nes the interpreter presented in
section 5. After loading the �le, the interpreter is started
by typing (boot-tower).

;;;
;;; A Simple Reflective Interpreter
;;;
;;;
;;; In some Scheme systems it may be necessary to
;;; first load a separate file that initializes
;;; every global variable in this file to some
;;; arbitrary value using DEFINE. It may also be
;;; necessary to include the following definition
;;; in that file:
;;;
;;; (define void
;;; (let ((g (cons '* '*)))
;;; (lambda () g)))
;;;

(set! evaluate
(lambda (e r k)
((if (constant? e)

evaluate-constant
(if (variable? e)
evaluate-variable
(if (if? e)
evaluate-if
(if (assignment? e)
evaluate-assignment
(if (abstraction? e)
evaluate-abstraction
evaluate-combination)))))

e r k)))

(set! evaluate-constant
(lambda (e r k)
(k (constant-part e))))

(set! evaluate-variable
(lambda (e r k)
(get-pair e r
(lambda (success-pair)
(k (cdr success-pair)))

(lambda ()
(wrong "symbol not bound: " e)))))

(set! wrong
(lambda (message object)
(display "Error: ")
(display message)
(display object)
(newline)))

(set! evaluate-if
(lambda (e r k)
(evaluate (test-part e) r
(lambda (v)
(if v
(evaluate (then-part e) r k)
(evaluate (else-part e) r k))))))

(set! evaluate-assignment
(lambda (e r k)

(evaluate (value-part e) r
(lambda (v)
(get-pair (id-part e) r
(lambda (success-pair)
(set-cdr! success-pair v)
(k (void)))

(lambda ()
(set-cdr! global-env

(cons (cons (id-part e) v)
(cdr global-env)))

(k (void))))))))

(set! evaluate-abstraction
(lambda (e r k)

(k (make-compound
(formals-part e) (body-part e) r))))

(set! evaluate-combination
(lambda (e r k)

(evaluate (operator-part e) r
(lambda (proc)
(if (reifier? proc)
((reifier-to-compound proc)
(operands-part e) r k)

(evaluate-operands (operands-part e) r
(lambda (args)

(apply-procedure proc args k))))))))

(set! evaluate-operands
(lambda (operands r k)

(if (null? operands)
(k '())
(evaluate (car operands) r
(lambda (v)
(evaluate-operands (cdr operands) r
(lambda (w)

(k (cons v w)))))))))

(set! evaluate-sequence
(lambda (body r k)

(if (null? (cdr body))
(evaluate (car body) r k)
(evaluate (car body) r
(lambda (v)
(evaluate-sequence (cdr body) r k))))))

(set! apply-procedure
(lambda (proc args k)

(if (compound? proc)
(evaluate-sequence
(procedure-body proc)
(extend
(procedure-environment proc)
(procedure-parameters proc)
args)

k)
(k (apply-primitive

(procedure-name proc) args)))))

(set! apply-primitive
(lambda (name args)
(if (eq? name 'car)

(car (1st args))
(if (eq? name 'cdr)

(cdr (1st args))
(if (eq? name 'cons)

(cons (1st args) (2nd args))
(if (eq? name 'set-car!)

(set-car! (1st args) (2nd args))
(if (eq? name 'set-cdr!)

(set-cdr! (1st args) (2nd args))
(if (eq? name 'assq)

(assq (1st args) (2nd args))
(if (eq? name 'memq)

(memq (1st args) (2nd args))
(if (eq? name 'null?)

(null? (1st args))
(if (eq? name '=)

(= (1st args) (2nd args))
(if (eq? name 'eq?)

(eq? (1st args) (2nd args))
(if (eq? name 'newline)

(newline)
(if (eq? name 'write)

(write (1st args))
(if (eq? name 'display)

(display (1st args))
(if (eq? name 'read)

(if (null? args) (read) (read (1st args)))
(if (eq? name '+)

(+ (1st args) (2nd args))
(if (eq? name '-)

(- (1st args) (2nd args))
(if (eq? name '*)

(* (1st args) (2nd args))
(if (eq? name 'symbol?)

(symbol? (1st args))
(if (eq? name 'list)

args
(if (eq? name 'pair?)

(pair? (1st args))
(if (eq? name 'eof-object?)

(eof-object? (1st args))
(if (eq? name 'close-input-port)

(close-input-port (1st args))
(if (eq? name 'open-input-file)

(open-input-file (1st args))
(if (eq? name 'void)

(void)
"Shouldn't Happen!"))))))))))))))))))))))))))

;;;
;;; Environments.
;;;

(set! extend
(lambda (r ids vals)
(if (null? ids)
r
(extend
(cons (cons (car ids) (car vals)) r)
(cdr ids)
(cdr vals)))))

(set! get-pair
(lambda (id r success failure)

(find-pair id r
success
(lambda ()
(find-pair
id global-env success failure)))))

(set! find-pair
(lambda (elt alist success failure)

((lambda (assq-result)
(if assq-result

(success assq-result)
(failure)))

(assq elt alist))))

(set! empty-env '())

;;;
;;; List utilities.
;;;

(set! 1st (lambda (l) (car l)))
(set! 2nd (lambda (l) (car (cdr l))))
(set! 3rd (lambda (l) (car (cdr (cdr l)))))
(set! 4th (lambda (l) (car (cdr (cdr (cdr l))))))

(set! test-tag
(lambda (tag)

(lambda (e)
(if (pair? e) (eq? (car e) tag) #f))))

;;;
;;; Procedures.
;;;

(set! make-compound
(lambda (formals body r)

(list 'compound formals body r)))

(set! compound? (test-tag 'compound))

(set! make-primitive
(lambda (op)

(list 'primitive op)))

(set! primitive? (test-tag 'primitive))

(set! primitive-identifiers
(lambda ()

'(car cdr cons set-car! set-cdr! assq memq
null? = eq? newline write display read
+ - * symbol? list pair? eof-object?
close-input-port open-input-file void)))

(set! make-reifier
(lambda (formals body r)

(list 'reifier formals body r)))

(set! reifier-to-compound
(lambda (reifier)

(cons 'compound (cdr reifier))))

(set! compound-to-reifier
(lambda (compound)
(cons 'reifier (cdr compound))))

(set! reifier? (test-tag 'reifier))

(set! procedure-parameters 2nd)
(set! procedure-body 3rd)
(set! procedure-environment 4th)
(set! procedure-name 2nd)

;;;
;;; Syntax.
;;;

(set! variable? symbol?)
(set! if? (test-tag 'if))
(set! assignment? (test-tag 'set!))
(set! abstraction? (test-tag 'lambda))
(set! quote? (test-tag 'quote))

(set! constant?
(lambda (e)
(if (pair? e) (quote? e)
(if (symbol? e) #f #t))))

(set! constant-part
(lambda (e) (if (quote? e) (2nd e) e)))

(set! test-part 2nd)
(set! then-part 3rd)
(set! else-part 4th)

(set! id-part 2nd)
(set! value-part 3rd)

(set! formals-part 2nd)
(set! body-part (lambda (e) (cdr (cdr e))))

(set! operator-part 1st)
(set! operands-part cdr)

;;;
;;; Read-Eval-Print Loop and Loadfile
;;;

(set! openloop
(lambda (read-prompt write-prompt)
(display read-prompt)
(evaluate (read) empty-env
(lambda (v)
(display write-prompt)
(if (eq? v (void))

"Nothing will be displayed"
(write v))

(newline)
(openloop read-prompt write-prompt)))))

(set! loadfile
(lambda (file)

((lambda(port)
((lambda (loop)

(set! loop
(lambda (v)

(if (eof-object? v)
(close-input-port port)
(evaluate v empty-env
(lambda (ignore)
(loop (read port)))))))

(loop (read port)))
'*))

(open-input-file file))))

;;;
;;; The booting process
;;;

(set! mapper
(lambda (f l)

(if (null? l)
'()
(cons (f (car l)) (mapper f (cdr l))))))

(set! initialize-global-env
(lambda ()

(set! global-env
(extend
empty-env
(primitive-identifiers)
(mapper make-primitive
(primitive-identifiers))))))

(set! boot-flat
(lambda ()

(initialize-global-env)
(openloop "0< " "0> ")))

(set! boot-tower
(lambda ()

(initialize-global-env)
(loadfile this-file-name)
(set-cdr! global-env
(cons (cons 'global-env global-env)

(cdr global-env)))
(openloop "0> " "0: ")))

(set! this-file-name "simple.ss")

